

# Introdução à Ciência da Computação

## Estruturas de Controle – Parte II

Prof. Ricardo J. G. B. Campello

### Créditos

Parte dos slides a seguir foram adaptados dos originais de A. L. V. Forbellone e H. F. Eberspächer

## Aula de Hoje

- Estruturas de Repetição
  - Teste no Início
  - Teste no Final
  - Variável de Controle

## Estruturas de Repetição

- São aquelas que permitem executar mais de uma vez (repetir) um determinado trecho do algoritmo
- O trecho do algoritmo em repetição é também chamado de laço (ou “loop”)
- As repetições devem ser sempre finitas
- Quanto à quantidade de repetições, os laços podem ser
  - Pré-determinados: Sabe-se antes a quantidade de execuções
  - Indeterminados: Não se conhece a quantidade de execuções
- Quanto ao critério de parada, os laços podem utilizar
  - Teste no início
  - Teste no final
  - Variável de controle

## Repetição com Teste no Início

- Laço que verifica antes de cada execução, se é “permitido” executar o trecho do algoritmo
- Trata-se de um laço que se mantém repetindo enquanto uma dada condição permanecer verdadeira

```
enquanto (condição) faça  
  comando 1;  
  comando 2;  
  ...  
  comando n;  
fimenquanto;
```

## Repetição com Teste no Início

- **Contador:**
  - Variável que reproduz o processo de contagem

<b>início</b>	
inteiro: CON;	
CON ← 0;	
<b>enquanto</b> (CON < 3) <b>faça</b>	
CON ← CON + 1;	
<b>fimenquanto</b> ;	
<b>fim.</b>	

CON
3

## Repetição com Teste no Início

Algoritmo 3.8 - Média Aritmética para 50 alunos

```
início  
  // declaração de variáveis  
  real: N1, N2, N3, N4, // notas bimestrais  
    MA; // média anual  
  inteiro: CON; // contador  
  CON ← 0; // inicialização do contador  
  enquanto (CON < 50) faça // teste da condição de parada  
    leia (N1, N2, N3, N4);  
    MA ← (N1 + N2 + N3 + N4) / 4;  
    escreva (MA);  
    se (MA >= 7) então  
      escreva ("Aluno Aprovado. Parabéns!");  
    senão  
      escreva ("Aluno Reprovado. Estude mais!");  
    fimse;  
    CON ← CON + 1; // incremento do contador  
  fimenquanto;  
fim.
```

## Repetição com Teste no Início

- **Acumulador:**

– Variável que reproduz o processo de acumulação

<b>início</b>
<b>inteiro</b> : CON, X, ACM;
CON ← 0;
ACM ← 0;
<b>enquanto</b> (CON < 3) <b>faça</b>
CON ← CON + 1;
<b>leia</b> (X);
ACM ← ACM + X;
<b>fimenquanto</b> ;
<b>fim.</b>

CON	ACM	X
3	11	4

## Repetição com Teste no Início

Algoritmo 3.9 - Média Aritmética da turma de 50 alunos

```
início  
  // declaração de variáveis  
  real: MA, // média anual de dado aluno  
        ACM, // Acumulador  
        MAT; // Média Anual da Turma  
  inteiro: CON; // contador  
  CON ← 0; // inicialização do contador  
  ACM ← 0; // inicialização do acumulador  
  enquanto (CON < 50) faça // teste da condição de parada  
    leia (MA);  
    ACM ← ACM + MA; // soma em ACM os valores lidos em MA  
    CON ← CON + 1; // incremento do contador  
  fimenquanto;  
  MAT ← ACM / 50; // calculo da média anual da turma  
  escreva ("média anual da turma =", MAT);  
fim.
```

## Repetição com Teste no Início

Exemplo:

- Faça um algoritmo que receba como entrada uma base e um expoente inteiros positivos e calcule o resultado da base elevada ao expoente, **sem** utilizar a operação de exponenciação (pot)

Possível  
Solução:

```
início  
  inteiro: B, E, R, // Base, Expoente e Resultado  
          C; // Contador  
  leia (B, E);  
  C ← 1;  
  R ← 1;  
  enquanto (C <= E) faça  
    R ← R*B;  
    C ← C + 1;  
  fimenquanto;  
  escreva ("pot(", B, " ", E, ")=", R);  
fim.
```

Funciona também para E = 0 !

## Repetição com Teste no Início

### Exercício:

- Faça um algoritmo que receba como entrada uma base e um expoente inteiros positivos e calcule o resultado da base elevada ao expoente, **sem** utilizar as operações de exponenciação (pot) e multiplicação (\*)

- **Dica**

Note que:

- $B = 1 + 1 + \dots + 1$  (B vezes)
- $B^2 = B + B + \dots + B$  (B vezes)
- $B^3 = B^2 + \dots + B^2$  (B vezes)
- $B^4 = B^3 + \dots + B^3$  (B vezes)
- ....
- $B^E = B^{(E-1)} + \dots + B^{(E-1)}$  (B vezes)

## Repetição com Teste no Final

- Laço que verifica, depois de cada execução, se é “permitido” continuar executando o trecho do algoritmo
- Trata-se de um laço que se mantém repetindo até que uma dada condição se torne verdadeira

### repita

comando 1;

comando 2;

...

comando n;

**até** (condição);

- Qual a principal diferença para o teste no início?

## Repetição com Teste no Final

Algoritmo 3.12 - Média Aritmética da turma com Repita

```
início  
  // declaração de variáveis  
  real: MA, // média anual de dado aluno  
        ACM, // Acumulador  
        MAT; // Média Anual da Turma  
  inteiro: CON; // contador  
  CON ← 0; // inicialização do contador  
  ACM ← 0; // inicialização do acumulador  
  repita  
    leia (MA);  
    ACM ← ACM + MA; // soma em ACM os valores lidos em MA  
    CON ← CON + 1; // incremento do contador  
  até (CON = 50); // teste da condição de parada  
  MAT ← ACM / 50; // calculo da média anual da turma  
  escreva ("média anual da turma = ", MAT);  
fim.
```

## Repetição com Teste no Final

Exemplo:

- Faça um algoritmo que receba como entrada dois números inteiros positivos e calcule o resto da divisão inteira desses números, **sem** utilizar as operações de divisão (/), multiplicação (\*), div e mod

Possível  
Solução:

```
início  
  inteiro: N, D, // Numerador, Denominador  
          R; // Resto da Divisão Inteira  
  leia (N, D);  
  R ← N;  
  repita  
    R ← R - D;  
  até (R < 0);  
  R ← R + D;  
  escreva ("Resto (" , N, " mod ", D, ") = ", R);  
fim.
```

## Repetição com Teste no Final

### Exercícios:

- Uma seqüência do tipo de Fibonacci é tal que, a partir de dois valores iniciais, cada valor seguinte é dado pela soma dos dois valores anteriores (e.g. 1 1 2 3 5 8 13 21 ... ). Faça um algoritmo que receba dois valores não-negativos iniciais e a quantidade de números da seqüência de Fibonacci a ser calculada e apresentada pelo algoritmo. Use repetição com teste no final
- Faça um algoritmo que repetidamente leia valores numéricos até que um valor negativo seja lido, e então apresente o máximo e o mínimo valores lidos, apresentado uma mensagem "Nenhum Valor !" caso o primeiro valor lido tenha sido negativo. Use repetição com teste no final
- Use repetição com teste no final para fazer um algoritmo que leia um número N inteiro não-negativo e então calcule e escreva o seu fatorial:
  - $N! = N \times (N-1) \times (N-2) \times \dots \times 1$  ou  $0! = 1$  (por definição)

## Repetição com Variável de Controle

- Laço simplificado para utilização em repetições de quantidade predeterminada
- Incorpora internamente o funcionamento de um contador de repetições

**para V de vi até vf passo p faça**

comando 1;

comando 2;

...

comando n;

**fimpara;**

## Repetição com Variável de Controle

Algoritmo 3.12 - Média Aritmética da turma com “para”

```
início  
  // declaração de variáveis  
  real: MA, // média anual de dado aluno  
        ACM, // Acumulador  
        MAT; // Média Anual da Turma  
  inteiro: V; // contador  
  ACM ← 0; // inicialização do acumulador  
  para V de 1 até 50 passo 1 faça  
    leia (MA);  
    ACM ← ACM + MA; // soma em ACM os valores lidos em MA  
  fimpara;  
  MAT ← ACM / 50; // calculo da média anual da turma  
  escreva (“média anual da turma = ”, MAT);  
fim.
```

## Repetição com Variável de Controle

Exemplo:

- Seja um número H dado por  $H = 1 + 1/2 + 1/3 + 1/4 + \dots + 1/N$ . Faça um algoritmo que leia  $N \geq 1$  e então calcule e escreva H.

```
início  
  inteiro: N, // Tamanho da Série  
          V; // Contador (var. de controle)  
  real: H; // Valor da Série  
  leia (N);  
  H ← 0;  
  para V de 1 até N passo 1 faça  
    H ← H + 1 / V; // Guarda o resultado a cada D contagens  
  fimpara;  
  escreva (“H = ”, H);  
fim.
```

## Repetição com Variável de Controle

### Exercícios:

- A conversão de graus Fahrenheit em Celsius é feita pela fórmula  $C = 5/9*(F - 32)$ . Escreva um algoritmo que calcule e escreva uma tabela de graus Celsius em função de Fahrenheit variando esta última grandeza de 50 a 150, de 1 em 1.
- Reescreva os exemplos e exercícios anteriores que utilizam repetições com teste no início e no final utilizando repetição com variável de controle. Se acreditar que algum deles não pode ser reescrito dessa maneira, justifique o porquê.

## Comparação entre Estruturas de Repetição

- Aprendemos 3 maneiras de construir laços de repetição
- É importante perceber que existem laços mais adequados ou convenientes para cada situação

Estrutura	Condição	Quantidade de Execuções	Condição de Repetição
Enquanto	Início	zero ou mais	Condição verdadeira
Repita	Final	uma ou mais	Condição falsa
Para	Não tem	$((vf - vi) \text{ div } p) + 1$	$v \leq vf (p > 0)$ ou $v \geq vf (p < 0)$

Nota: p = tamanho do passo (inteiro não nulo)

## Exercícios Adicionais

- ◆ Capítulo III de (Forbellone & Eberspächer, 2005):
  - Estudar exercícios de fixação
  - Resolver exercícios propostos